
Abstract

We illustrate the use of the computer algebra system Macaulay2

for simplifications of classical unirationality proofs. We explicitly treat

the moduli spaces of curves of genus g=10, 12 and 14.

If a moduli space M is unirational, then ultimately we would like to ex-
hibit explicitly a dominating family, which depends only on free parameters.
Although this is possible along the lines of a known unirationality proof in
principle, this is far beyond what computer algebra can do today in most
cases.

However, if we replace each generic free parameter with a random choice
of an element in a ground field F, then in many cases the computation of a
random element of the family defined over F is possible.

This is particular interesting over a finite field F, because in that case
there is no growth of the coefficients in Gröbner basis computations. The
equidistributed probability distribution on F induces a probability measure
onM(F) via the algorithm, and opens up for us the possibility to investigate
the moduli space experimentally. With semi-continuity arguments we can
sometimes deduce effectiveness of certain divisors on the moduli space. The
computation with a single random example over a finite field can verify that
the constructed family indeed dominates the desired moduli space. The key
advantage in using Computer algebra instead of theoretical arguments is that
random choices give smooth varieties almost certainly, while smoothness is
always a difficult issue in any theoretical treatment.

In this note we illustrate this technique in a number of cases, and provide
explicit code for the computer algebra system Macaulay2. A more elaborate

code is provided with the Macaulay2 package RandomCurves, which will be
freely available from [M2] and [BGS]. I recommend running the code available
from www.math.uni-sb.de/ag/schreyer/home/forHandbook.m2 parallel with
reading the code in the article. My favorite set-up of Macaulay2 is within
emacs, which for example has syntax highlighting. That makes the code
much more readable. Instructions how to use Macaulay2 with emacs come
with the installation package of Macaulay2, see [M2].
Acknowledgements. I thank the referees for their careful suggestions, and
Florian Geiß and Hans-Christian Graf von Bothmer for valuable discussions.

1 Random Plane Curves

Let Mg denote the moduli space of curves of genus g. For a general smooth
projective curve C of genus g the Brill-Noether locus

W r
d (C) = {L ∈ Picd(C) | h0(L) ≥ r + 1}

is non-empty and smooth away from W r+1
d (C) of dimension ρ if and only if

the Brill-Noether number

ρ = ρ(g, d, r) = g − (r + 1)(g − d+ r) ≥ 0,

equivalently, iff
d ≥ ((r + 1)(g − r)− g)/(r + 1).

Moreover, W r
d (C) is connected in case ρ > 0, see [ACGH]. The tangent space

at the point L ∈ W r
d (C) \W

r+1
d (C) is naturally dual to the cokernel of the

Petri map
H0(C,L)⊗H0(C, ωC ⊗ L

−1)→ H0(C, ωC).

To prove the unirationality of Mg we might prove that the Hilbert scheme
of the corresponding models of C in Pr is unirational.

Severi’s proof of the unirationality of Mg for g ≤ 10 is based on the fact
that a general curve up to genus 10 has a plane model with double points
in general position, see [Sev21, AS79]. An easy dimension count shows that
this cannot be true for g ≥ 11.

The algorithm that computes a random curve of genus g = 10 proceeds
in four steps.

1. Compute the minimal degree d such that ρ(g, 2, d) ≥ 0.

2. Choose a scheme ∆ of δ =
(
d−1
2

)
− g distinct points in P2.

3. Choose a random element in f ∈ H0(P2, I2∆(d)).

4. Certify that f defines an absolutely irreducible δ-nodal curve.

In case g = 10 we have d = 9 and δ = 18.
The Macaulay2 code for these steps looks at follows. The code is, I

believe, fairly self explaining, since the Macaulay2 language is close to stan-
dard mathematical notation. For further explanations I refer to the online
documentation http://www.math.uiuc.edu/Macaulay2/.

Step 1.

i1 : -- function that computes the minimal d:

dmin=(r,g)->ceiling(((r+1)*(g+r)-g)/(r+1))

i2 : g=10,r=2,d=dmin(r,g),delta=binomial(d-1,2)-g

i3 : rho=g-(r+1)*(g+r-d)

o3 = 1

Step 2. We specify δ points in P2:

i4 : FF=ZZ/10007 -- a finite ground field

i5 : S=FF[x_0..x_2] -- the homogeneous coordinate ring of P2

i6 : -- pick a list of delta ideals of random points:

points=apply(delta,i->ideal(random(1,S),random(1,S)))

Remark: The Macaulay2 function "random(n,S)" picks a random element
of degree n in a graded ring S.

i7 : IDelta=intersect points

i8 : betti res IDelta

0 1 2

o8 = total: 1 4 3

0: 1 . .

1: . . .

2: . . .

3: . . .

4: . 3 .

5: . 1 3

o8 = BettiTally

For a very small finite field we cannot find δ distinct points defined over F.
Even for p = 101 the procedure will not choose distinct points in P2(F101)
in about 1.5% of all cases. For p as large as 10007 the failure probability is
neglectable.

A method to get points which works for small finite fields is the follow-
ing: Since points have codimension 2 in P2, we might use the Hilbert-Burch
matrix, see [Ei95], Theorem 20.15. We recall the Hilbert-Burch theorem in
its most useful special case:

Theorem 1.1. Let R be a local (or graded) regular noetherian ring. The
minimal free resolution of R/I for a Cohen-Macaulay ideal I ⊂ R of codi-
mension 2 has shape

0← R/I ← R← Rn+1 ← Rn ← 0

where the map R← Rn+1 is given by the n+1 maximal minors of the matrix
Rn+1 ← Rn. Conversely, given a matrix (aij)i=0,...,n,j=1,...,n, whose maximal
minors have no common factor, then the ideal generated by the maximal
minors is Cohen-Macaulay of codimension 2 and the corresponding complex
is exact.

Note that (m0, . . . , mn)(aij) = 0 for mk = (−1)k det(aij)i 6=k follows by ex-
panding the determinants

0 = det



a01 . . . a0n a0i
...

...
...

an1 . . . ann ani




of matrices with a repeated column with respect to the last column. The
Hilbert-Burch Theorem is the basic reason why the Hilbert scheme of finite
length subschemes on a smooth surface is smooth, see [Fog68].

We continue with our Macaulay2 -program

i9 : M=random(S^{3:-4,1:-5},S^{3:-6}); -- a Hilbert-Burch matrix;

i10 : betti M

i11 : IDelta=minors(3,M); -- and its minors

i12 : betti res IDelta

0 1 2

o12 = total: 1 4 3

0: 1 . .

1: . . .

2: . . .

3: . . .

4: . 3 .

5: . 1 3

o12 : BettiTally

i13 : degree IDelta==delta

o13 = true

Step 3. Compute curves with double points in Delta:

i14 : -- the saturation of the ideal IDelta^2 contains

-- all equations which vanish double at Delta:

J=saturate(IDelta^2);

i15 : betti J

0 1

o15 = total: 1 10

0: 1 .

1: . .

2: . .

3: . .

4: . .

5: . .

6: . .

7: . .

8: . 1

9: . 9

o15 : BettiTally

As expected, there is precisely one curve with multiplicity 2 at every point
of ∆.

i16 : IC=ideal(gens J*random(source gens J, S^{-d}))

i17 : degree IC

o17 = 9

Step 4. To certify that we have indeed obtained a δ-nodal curve C, it
suffices to prove that the singular locus is reduced of degree δ because only
plane curves with at most nodes have a reduced singular locus.

i18 : singC=ideal jacobian IC + IC;

i19 : codim singC == 2 and degree singC == delta

o19 = true

i20 : -- remove primary component at the irrelevant ideal:

singCs=saturate(singC);

i21 : betti singCs, betti singC

0 1 0 1

o21 = (total: 1 4, total: 1 3)

0: 1 . 0: 1 .

1: . . 1: . .

2: . . 2: . .

3: . . 3: . .

4: . 3 4: . .

5: . 1 5: . .

6: . .

7: . 3

o21 : Sequence

i22 : codim (minors(2,jacobian singCs)+singCs) == 3

o22 = true

We verify that C is geometrically irreducible from our information so far.
Indeed, if the curve decomposes as C1∪C2 with (degC1, degC2) = (a, b), say
a < b, and a + b = 9, then the intersection points C1 ∩ C2 form a subset of
∆. So (a, b) = (1, 8) or (2, 7) is excluded because I∆ is generated in degree 6.
The case (4, 5) is excluded since 20 > 18 = δ and (3, 6) is excluded because
∆ is not a complete intersection.

Finally, we conclude from this computation that M10 is unirational over Q.
We first note that the computations in the finite prime field F may be viewed
as the reduction mod p of the analogous computations for a curve defined
over an open part of SpecZ. By semi-continuity, the curve over Q is δ-nodal
as well, hence has geometric genus 10, which proves that we have a rational
map An

99K M10 defined over Q for a suitable n. More precisely, we have a

correspondence

Hilbδ(P
2) H = {(C ′,∆) | C ′ ∈ PH0(P2, I2∆(d))

∗ is δ-nodal }oo

��
Mg

For g = 10 the left arrow gives a birational map of a component H ′ of H
onto a dense open subset of Hilbδ(P

2), because χ(I2∆(9)) =
(
11
2

)
− 3δ = 1,

and h0(I2∆(9)) = 1 holds for our specific point ∆ ∈ Hilbδ(P
2).

The downward arrow factors through the universal W2
d ⊂ Pic

d
g → Mg,

which has codimension at most (2 + 1)(g + 2 − d) in the universal Picard
variety over Mg at every point. A fiber of H →W2

d over a point (C,L) with
h0(C,L) = 3 is precisely the PGL(3) orbit. Thus to prove that H ′ → M10

dominates, it remains to certify that the fiber ofW2
d →M10 over our specific

point C has expected dimension ρ at the specific line bundle L = η∗OP2(1) ∈
W 2

d (C) where η : C → C ′ ⊂ P2 denotes the normalization map.
By Riemann-Roch and adjunction h0(C,L) = r + 1 = 3 holds because

h1(C,L) = h0(C, ωC ⊗ L
−1) = h0(P2, I∆(d − 4)) = 3. Moreover, the Petri

map can be identified with

H0(P2, I∆(d− 4))⊗H0(P2,O(1))→ H0(P2, I∆(d− 3))

which is an injection, since there is no linear relation among the three quintic
generators of I∆ by the shape of the Hilbert-Burch matrix, compare also with
the output line o8 and o12 of the Macaulay2 program above. Thus W 2

d (C)
is actually smooth of dimension ρ in L as desired.

It is easy to transform the code above into a function which chooses
randomly δ-nodal curves of degree d provided the expected h0(P2, I∆(d)) =(
d+2
2

)
− 3δ > 0. An implementation can be found in the Macaulay2 package

RandomCurves.

2 Searching

Over a finite field F we might find points C ∈ M(F) of a moduli space,
provided M is dominated by a variety H of fairly low codimension in a uni-
rational variety G. Indeed, if H is absolutely irreducible, then the proportion

of F-rational points is approximately

| H(F) |

| G(F) |
∼ q−c

where q =| F | and c is the codimension of H ⊂ G by the Weil formulas. If
we can decide p /∈ H fast enough computationally, then we might be able to
find points in H(Fq) for small q by picking points at random in G(Fq) and
testing p ∈ H . I will illustrate this technique by searching for plane models
of random curves of genus 11. Of course, to get just a random curve of genus
11, it is much better to use Chang and Ran’s unirational parameterization
of M11 via space curves, as indicated in the next section.

This time we will use a bit more of the Macaulay2 syntax. In particular
we will illustrate the use of method functions with options.

i23: randomDistinctPlanePoints = method(TypicalValue=> Ideal);

-- create the ideal of k random points

-- via a Hilbert-Burch matrix

i24: randomDistinctPlanePoints (ZZ,PolynomialRing) := (k,S) -> (

if dim S =!= 3 then error "no polynomial ring in 3 variables";

-- numerical data for the Hilbert-Burch matrix

n := ceiling((-3+sqrt(9.0+8*k))/2);

eps := k - binomial(n+1,2);

a := n+1-eps;

b := n-2*eps;

distinct := false;

while not distinct do (

-- the Hilbert-Burch matrix is B

B := if b >= 0 then random(S^a,S^{b:-1,eps:-2})

else random(S^{a:0,-b:-1}, S^{eps:-2});

I := minors(rank source B, B);

distinct = distinctPlanePoints I);

return I);

i25: distinctPlanePoints=method(TypicalValue=>Boolean);

i26: distinctPlanePoints(Ideal):= I-> (

dim I==1 and dim (I+minors(2,jacobian I))<=0)

i27: distinctPlanePoints(List):= L->(

-- for a List of ideals of points

-- check whether they have some point in common.

degree intersect L == sum(L,I->degree I))

The function randomDistinctPlanePoints will return an ideal of a set of
distinct points which we use in our search below.

The minimal degree of plane models of a general curve of genus g = 11
is d = 10, and we expect that a general model will have δ =

(
d−1
2

)
− 11 = 25

ordinary double points. Since χ(P2, I2∆(10)) = −9 we expect that ∆ with
h0(P2, I2∆(10)) = 1 from a codimension 10 subfamily of Hilb25(P

2). We can
improve our odds if we look for models with 2 triple points. Since triple points
occur in codimension 1 and ρ(11, 2, 10) = 2 we expect that the family of plane
curves of degree 10 with two triple points p1, p2 and 19 double points will
dominate M11. Since χ(P

2, I2∆⊗I
3
p1
⊗I3p2(10)) =

(
10+2
2

)
− 3 · 19− 6 · 2 = −3,

we expect a search for points in a subfamily of codimension 4. Thus the
search function below will have an average running time of order O(q4) with
respect to the number of elements of F = Fq. So this will be only feasible for
very small finite fields.

i28: searchPlaneGenus11Curve=method(Options=>{Attempts=>infinity})

-- search a plane curve of degree 10,geometric genus 11 with

-- two triple points and 19 double points

i29: searchPlaneGenus11Curve PolynomialRing := opt -> S -> (

I1 := ideal(S_0,S_1);

I2 := ideal(S_1,S_2);

IDelta := ideal 0_S; J := ideal 0_S;

h := -3; attempt := 0;

while h <= 0 and attempt < opt.Attempts do (

IDelta = randomDistinctPlanePoints(19,S);

while not distinctPlanePoints({I1,I2,IDelta}) do (

IDelta = randomDistinctPlanePoints(19,S));

J = intersect(I1^3,I2^3, saturate(IDelta^2));

h = (tally (degrees gens truncate(10,J))_1)_{10};

attempt = attempt +1);

print attempt;

if attempt >= opt.Attempts then return null;

f := ideal 0_S;

gJ := gens J;

if h === 1 then f = ideal gJ_{0}

else while f == 0 do f=ideal(gJ*random(source gJ, S^{ -10}));

singf := ideal singularLocus f;

doublePoints := saturate(singf, I1*I2);

if degree doublePoints == 19 and (

f31 := ideal contract(S_2^7, gens f);

dim singularLocus f31 == 1)

and (

f32 := ideal contract(S_0^7, gens f);

dim singularLocus f32 == 1)

then return f else

searchPlaneGenus11Curve(S, Attempts => opt.Attempts-attempt));

After these preparations the commands below will return a desired curve
within a few minutes.

i30 : p=5;FF=ZZ/p -- a finite ground field

i32 : S=FF[x_0..x_2]

i33 : setRandomSeed("alpha")

i34 : C=searchPlaneGenus11Curve(S,Attempts=>2*p^4)

18

432

-- used 48.132 seconds

o34 : Ideal of S

3 Space Curves

The proof of the unirationality of the moduli space Mg for g = 11, 12 and
13 by Sernesi and Chang-Ran is based on models of these curves in P3, see
[Ser81, CR84].

Suppose C ⊂ P3 is a Cohen-Macaulay curve with ideal sheaf IC . The
Hartshorne-Rao module

M =
∑

n∈Z

H1(P3, IC(n)),

which has finite length and measures the deviation for C to be projectively
normal, plays an important role in liaison theory of curves in P3. We briefly
recall the basic facts.

Let S = F[x0, . . . , x3] and SC = S/IC denote the homogeneous coordinate
ring of P3 and C ⊂ P3 respectively. By the Auslander-Buchsbaum-Serre

formula, [Ei95] Theorem 19.9, SC has projective dimension pdS SC ≤ 3 as an
S-module. Thus the minimal free resolution has shape

0← SC ← S ← F1 ← F2 ← F3 ← 0

with free graded modules Fi = ⊕S(−j)
βij .

The sheafified kernel G = ker(F̃1 → OP3) is always a vector bundle, and

0← OC ← OP3 ← ⊕jOP3(−j)β1j ← G ← 0

is a resolution by locally free sheaves. If C is arithmetically Cohen-Macaulay,
then F3 = 0 and G splits into a direct sum of line bundles. In this case the
ideal IC is generated by the maximal minors of F1 ← F2 by the Hilbert-Burch
Theorem 1.1. In general we haveM ∼=

∑
n∈ZH

2(P3,G(n)) and
∑

n∈ZH
1(P3,G(n)) =

0.
The importance of M for liaison comes about as follows. Suppose f, g ∈

IC are homogeneous forms of degree d and e without common factor. Let
X = V (f, g) denote the corresponding complete intersection, and let C ′ be
the residual scheme defined by the homogeneous ideal IC′ = (f, g) : IC .
The locally free resolutions of OC and OC′ are closely related: Applying
Ext2(−, ωP3) to the sequence

0→ IC/X → OX → OC → 0

gives
0← Ext2(IC/X , ωP3)← ωX ← ωC ← 0.

From ωX
∼= OX(d+ e− 4) we conclude Ext2(IC/X ,OP3(−d− e)) ∼= OC′ and

hence IC′/X
∼= ωC(−d− e+ 4).

The mapping cone

0 OC
oo Ooo

⊕

j

O(−j)β1j
oo Goo 0oo

0 OX
oo

OO

Ooo

∼=

OO

O(−d)⊕O(−e)oo

OO

O(−d− e)oo

OO

0oo

dualized with Hom(−,O(−d − e)) yields the locally free resolution

0→
⊕

j

O(j − d− e)β1j → G∗(−d − e)⊕O(−e)⊕O(−d)→ O → OC′ → 0

11



In particular one has

MC′ =
∑

n∈Z

H1(P3, IC′(n)) ∼=
∑

n∈Z

H1(P3,G∗(n− d− e))

∼=
∑

n∈Z

H2(P3,G(d+ e− 4− n))∗ ∼= HomF(MC ,F)(4− d− e)

Thus curves, which are related via an even number of liaison steps, have the
same Hartshorne-Rao module up to a twist. Rao’s famous result [Rao78]
says that the even liaison classes are in bijection with finite length graded
S-modules up to twist.

Reversing the role of C and C ′, we conclude that the ideal sheaf of C has
a locally free resolution

0← IC ← F ⊕ L1 ← L2 ← 0

where L1 = ⊕ℓO(−cℓ) and L2 = ⊕kO(−dk) are direct sums of line bundles,
while F is a locally free sheaf without line bundle summands. Note that F
has no H2-cohomology, and its H1-cohomology

∑

n∈Z

H1(P3,F(n)) ∼= M

is the Hartshorne-Rao module of C. The map L2 → L1 coming from a liaison
construction might be non-minimal, in which case one can cancel summands
of L1 against some summands of L2. Below we will work with complexes
which arise after such cancellation.

Since there are no line bundle summands and no H2, F is determined by
its H1-cohomology: Consider a minimal presentation of M

0←M ← ⊕iS(−ai)← ⊕jS(−bj)← N ← 0

and the kernel N . Then F ∼= Ñ is the associated coherent sheaf of N .
The main difficulty in constructing space curves lies in the construction

of M . Given M we can find the curve as the cokernel of an homomorphism
ϕ ∈ Hom(L2,F ⊕ L1), which we may regard as a vector bundle version of
a Hilbert-Burch matrix. Frequently in interesting examples we will have
L1 = 0.

We apply this approach for the case g = 12 and d = 13.

To get an idea about the twists in the resolutions of M and C, we use
the Hilbert numerators: Suppose the minimal finite free resolution of M is

0←M ✛ F0 ← F1 ← . . .← F4 ← 0

with
Fi = ⊕S(−j)

βij

then the Hilbert series

HM(t) =
∑

n

dimMnt
n =

∑
i,j(−1)

iβijt
j

(1− t)4
.

Assuming the open condition that C has maximal rank, i.e.

H0(P3,O(n))→ H0(C,Ln)

is of maximal rank for all n then HM(t) = 5t2 + 8t3 + 6t4 and the Hilbert
numerator is

HM(t)(1− t)4 = 5t2 − 12t3 + 4t4 + 4t5 + 9t6 − 16t10 + 6t11

If M has a natural resolution, which means that for each j at most one βij
is non-zero, then M has a Betti table (βi,i+j)

0 1 2 3 4

total: 5 12 17 16 6

2: 5 12 4 . .

3: . . 4 . .

4: . . 9 16 6

in Macaulay2 notation. Having natural resolution is another open condition.
If SC = S/IC has natural resolution as well, then its Betti table will be

0 1 2 3

total: 1 11 16 6

0: 1 . . .

1: . . . .

2: . . . .

3: . . . .

4: . 2 . .

5: . 9 16 6

13



Comparing these tables, we conclude that L1 = 0, rankF = 12− 5 = 7 and
L2 = O(−4)

4⊕O(−5)2 with rankL2 = 6 will hold for an open set of curves.
Of course, we have to prove that the set is non-empty and contains smooth
curves.

To construct a desiredM , we start with the submatrix ψ defining S12(−3)←S4(−4)
in F1 ← F2 which we choose randomly. Since 12

(
3+1
3

)
− 4

(
3+2
3

)
= 8 > 5, the

kernel ker(ψt : S12(3)→ S4(4)) has at least 8 linearly independent elements
of degree −2, and we choose 5 random linear combinations of these elements
to get the transpose of the presentation matrix F0 ← F1 of M .

i35 : FF=ZZ/10007

i36 : S=FF[x_0..x_3]

i37 : psi=random(S^{12:-3},S^{4:-4}) -- the submatrix

i38 : betti(syzpsit=syz transpose psi)

i39 : M=coker transpose(syzpsit*random(source syzpsit,S^{5:2}));

i40 : F= res M -- free resolution of the desired Hartshorne-Rao module

i41 : betti F

i42 : L2=S^{4:-4,2:-5}

i43 : phi= F.dd_2*random(F_2,L2);

i44 : betti(syzphit=syz transpose phi)

i45 : IC=ideal mingens ideal( transpose syzphit_{5}*F.dd_2);

i46 : betti res IC -- free resolution of C

i47 : codim IC, degree IC, genus IC

o47 = (2, 13, 12)

o47 : Sequence

Next we check smoothness. Since C ⊂ P3 has codimension 2 and is locally
determinantal, it is unmixed, and we can apply the jacobian criterion.

i48 : singC=IC+minors(2,jacobian IC);

i49 : codim singC==4

o49 = true

Thus the curve C is smooth. We conclude that the Hilbert scheme Hilbd,g(P
3)

of curves of degree d = 13 and g = 12 has a unirational component H defined
over Q, whose general element is smooth.

As a corollary we get the unirationality of M12 from this if we verify
that the fiber of the rational map H 99K M12 has the expected dimension
dimPGL(4) + ρ(g, r, d) = 15 + 4 at the given point. Now this holds if

the Petri map H0(ωC(−1)) ⊗ H
0(O(1)) → H0(ωC) is injective where ωC =

Ext2(OC , ωP3) ∼= Ext1(IC ,OP3(−4)) denotes the canonical bundle on C.

i50 : betti Ext^1(IC,S^{-4})

0 1

o50 = total: 6 12

-1: 2 .

0: 4 12

shows that there are no linear relation among the two generators of

Γ∗(ωC) =
∑

n∈Z

H0(P3, ωC(n)) = Ext1S(IC , S(−4))

in degree −1. Thus H dominates and M12 is unirational. As a corollary of
our construction we obtain that the Hurwitz scheme Hk,g of k-gonal curves
of genus g is unirational for (k, g) = (9, 12). Indeed ωC(−1) is a line bundle
of degree k = 22− 13 = 9.

The case d = 13, g = 12 is actually not used in Sernesi’s proof of
the unirationality of M12. Chang, Ran and Sernesi use the cases (d, g) =
(10, 11), (12, 12) and (13, 13), all of which can be treated similarly as the
case above. I took the case (d, g) = (13, 12) because it illustrates the dif-
ficulty in constructing M well, and because being different, it implies some
minor new results. Computer algebra simplifies the cumbersome proof that
the approach leads to smooth curves in all of theses cases. For more details
and an implementation we refer to the Macaulay2 package RandomCurves.

There are 65 values for (d, g) such that there are possibly non-degenerate
maximal rank curves with natural resolution such that the Hartshorne-Rao
module has diameter ≤ 3 and natural resolution as well. For 60 of these
values one can establish the existence of a unirational component in the
Hilbert scheme by the methods above.

To prove the unirationality of Mg for g ≥ 14 via space curves, leads to
Hartshorne-Rao modules which have diameter ≥ 4, i.e., modules, which are
nonzero in at least 4 different degrees. The construction of such modules is
substantially more difficult.

4 Verra’s proof of the unrationality of M14

Verra’s idea [Ve05] is of beautiful simplicity. Consider a general curve of
genus g = 14 and a pencil |D| of minimal degree which is degD = 8. The

Serre dual linear system |K − D| embeds C into P6 as curve of degree 18
with expected syzygies

0 1 2 3 4 5

total: 1 13 45 56 25 2

0: 1 . . . . .

1: . 5 . . . .

2: . 8 45 56 25 .

3: . . . . . 2

In particular, C ⊂ P6 lies on 5 quadrics. The intersection of the quadrics
should have codimension 5 and degree 25 = 32, thus should equal C ∪ C ′

where the residual curve C ′ has smaller degree 32 − 18 = 14 than C and
hence also smaller genus, g(C ′) = 8 as it turns out.

(Indeed the dualizing sheaf of the complete intersection is ωC∪C′
∼= OC∪C′(5·

2−7) ∼= OC∪C′(3), and we obtain arithmetic genus pa(C∪C
′) = 25 ·3/2+1 =

49. Assuming only nodes as intersection points we have ωC∪C′ ⊗ OC
∼=

ωC(C ∩ C
′) and get deg(C ∩ C ′) = 3 · 18 − 26 = 28 intersection points.

The formula pa(C ∪ C
′) = pa(C) + pa(C

′) + deg(C ∩ C ′) − 1 finally gives
pa(C

′) = 49 − 14 − 28 + 1 = 8.) There is no reason to expect anything else
than, that C ′ is a smooth curve.

By a famous result by Mukai [Mu92], every general canonical curve of
genus 8 is obtained as transversal intersection

C ′ = G(2, 6) ∩ P7 ⊂ P14

of the Grassmannian G(2, 6) ⊂ P14 in its Plücker embedding. If we choose
8 general points on G(2, 6) and P7 as their span, then we get a genus 8
curve together with 8 points. We group these 8 points into two divisors
D1 = p1+ . . .+p4 and D2 = p5+ . . .+p8 of degree 4. Then | KC′ +D1−D2 |
is a general linear system of degree 14 on C ′, and re-embedding C ′ →֒ P6

leads to a curve with expected syzygies

0 1 2 3 4 5

total: 1 7 35 56 35 8

0: 1 . . . . .

1: . 7 . . . .

2: . . 35 56 35 8

The Macaulay2 code is now straightforward. First we construct C ′ in its
canonical embedding.

i51 : randomCurveOfGenus8With8Points = R ->(

--Input: R a polynomial ring in 8 variables,

--Output: a pair of an ideal of a canonical curve C

-- together with a list of ideals of 8 points

--Method: Mukai’s structure theorem on genus 8 curves.

-- Note that the curves have general Clifford index.

FF:=coefficientRing R;

p:=symbol p;

-- coordinate ring of the Plucker space:

P:=FF[flatten apply(6,j->apply(j,i->p_(i,j)))];

skewMatrix:=matrix apply(6,i->apply(6,j->if i<j then

p_(i,j) else if i>j then -p_(j,i) else 0_P));

-- ideal of the Grassmannian G(2,6):

IGrass:=pfaffians(4,skewMatrix);

points:=apply(8,k->exteriorPower(2,random(P^2,P^6)));

ideals:=apply(points,pt->ideal( vars P*(syz pt**P^{-1})));

-- linear span of the points:

L:= ideal (gens intersect ideals)_{0..6};

phi:=vars P%L; -- coordinates as function on the span

-- actually the last 8 coordinates represent a basis

phi2:= matrix{toList(7:0_R)}|vars R;

-- matrix for map from R to P/IC

IC:=ideal (gens IGrass%L); --the ideal of C on the span

-- obtained as the reduction of the Grassmann equation mod L

IC2:=ideal mingens substitute(IC,phi2);

idealsOfPts:=apply(ideals,Ipt->

ideal mingens ideal sub(gens Ipt%L,phi2));

(IC2,idealsOfPts))

Building upon Mukai’s result, we can construct the desired curve C ′:

i52 : randomNormalCurveOfGenus8AndDegree14 = S -> (

-- Input: S coordinate ring of P^6

-- Output: ideal of a curve in P^6

x:=symbol x;

FF:=coefficientRing S

R:=FF[x_0..x_7];

(I,points):=randomCurveOfGenus8With8Points(R);

D1:=intersect apply(4,i->points_i); -- divisors of degree 4

17



D2:=intersect apply(4,i->points_(4+i));

-- compute the complete linear system |K+D1-D2|, note K=H1

H1:=gens D1*random(source gens D1,R^{-1});

E1:=(I+ideal H1):D1; -- the residual divisor

L:=mingens ideal(gens intersect(E1,D2)%I);

-- the complete linear system

-- note: all generators of the intersection have degree 2.

RI:=R/I; -- coordinate ring of C’ in P^7

phi:=map(RI,S,substitute(L,RI));

ideal mingens ker phi)

i53 : FF=ZZ/10007;S=FF[x_0..x_6];

i55 : I=randomNormalCurveOfGenus8AndDegree14(S);

i56 : betti res I

Finally, we get curves of genus 14 with

i57: randomCurveOfGenus14=method(TypicalValue=>Ideal,

Options=>{Certify=>false})

-- The default value of the option Certify is false, because

-- certifying smoothness is expensive

i58 : randomCurveOfGenus14(PolynomialRing) :=opt ->( S-> (

-- Input: S PolynomialRing in 7 variables

-- Output: ideal of a curve of genus 14

-- Method: Verra’s proof of the unirationality of M_14

IC’:=randomNormalCurveOfGenus8AndDegree14(S);

-- Choose a complete intersection:

CI:=ideal (gens IC’*random(source gens IC’,S^{5:-2}));

IC:=CI:IC’; -- the desired residual curve

if not opt.Certified then return IC;

if not (degree IC ==18 and codim IC == 5 and genus IC ==14)

then return nil;

someMinors :=minors(5, jacobian CI);

singCI:=CI+someMinors;

if not (degree singCI==28 and codim singCI==6)

then return nulll;

someMoreMinors:=minors(5, jacobian (gens IC)_{0..3,5});

singC:=singCI+someMoreMinors;

if codim singC == 7 then return IC else return nil))

i59 : time betti(J=randomCurveOfGenus14(S))

i60 : time betti(J=randomCurveOfGenus14(S,Certified=>true))

i61 : betti res J

To deduce from these computation the unirationality of M14, we have to
prove again that the Petri map is injective. By the Betti numbers of the
free resolution of SC we see that there is is no linear relation among the two
generators of ωC(−1). Thus the family is dominant because the conditions
that C has expected Betti numbers and that C ′ is smooth are open.

Finally, we remark that, comparing the syzygies of OC , OC∪C′ and OC′

via liaison as outlined for space curves in Section 3, we see that the Petri
map of (C,OC(1)) is injective, if and only if IC′ is generated by quadrics.
Indeed, suppose the Betti table of C ′ is

0 1 2 3 4 5

0: 1 . . . . .

1: . 7 x . . .

2: . x 35 56 35 8

where we assume that we need x cubic generators of the ideal of IC′. Then
the Betti table of the mapping cone with the Koszul complex resolving OC∪C′

is

0 1 2 3 4 5 6

-1: . 1 . . . . .

0: 1 . 5 . . . .

1: . 7 x 10 . . .

2: . x 35 56 45 8 .

3: . . . . . 5 .

4: . . . . . . 1

Minimalizing the dual complex, leads to following Betti table of C:

0 1 2 3 4 5

0: 1 . . . . .

1: . 5 . . . .

2: . 8 45 56 25 x

3: . . . . x 2

5 Minimal Resolution Conjectures and Koszul

Divisors

A graded S-module M is said to satisfy the minimal resolution conjecture
(MRC) (or is said to have expected syzygies) if the Betti numbers βij of the
minimal free resolution

0 ✛ M ✛ F0
✛ F1

✛ . . . Fc
✛ 0

where Fi = ⊕jS(−j)
βij satisfy the following: For each internal degree j at

most one of the numbers βij 6= 0. In other words, M satisfies MRC if one
can nearly read off the Betti table from the Hilbert numerator of the Hilbert
series HM .

We say that a module M has a pure resolution, if for each homological
degree i at most one βij 6= 0. Betti tables of Cohen-Macaulay modules with
pure resolutions play a special role since they span the extremal rays in the
Boij-Söderberg cone of all Betti tables, see [ES09, BS08]. Moreover, ifM has
a pure resolution, then M satisfies the MRC, see [ES09].

We say that the MRC holds generically on a component H of a Hilbert
scheme Hilbp(t)(P

n) if it is satisfied for the coordinate ring SX for X ∈ U ⊂
H ⊂ Hilbp(t)(P

n) in a dense open subset U of H . Note that the Hilbert
function is constant on an open set U ′ of H , so this makes sense. Since Betti
numbers are upper semi-continuous in flat families with constant Hilbert
function, there is a smallest possible Betti table for X ∈ U ′, and we can ask
what this table is. If the MRC is satisfied generically on H , then in a sense,
the question, what is the generic Betti table, has a trivial answer.

The MRC has been established in various cases, the most famous one is
the following.

Theorem 5.1 (Voisin, 2005 [Vo02], Generic Green’s Conjecture [Gr84]). A
general canonical curve of genus g ≥ 3 over a ground field of characteristic
0 satisfies the MRC.

In more concrete terms, this conjecture says for example, that a general
canonical curve of genus g = 15 has the Betti table

0 1 2 3 4 5 6 7 8 9 10 11 12 13

total: 1 78 560 2002 4368 6006 4576 4576 6006 4368 2002 560 78 1

0: 1 . . . . . . . . . . . . .

1: . 78 560 2002 4368 6006 4576 . . . . . . .

2: . . . . . . . 4576 6006 4368 2002 560 78 .

3: . . . . . . . . . . . . . 1

Note that the symmetry of the table comes from the fact that the homo-
geneous coordinate ring of a canonical curve is Gorenstein. Before Voisin’s
result, g = 15 was for a long time the bound of how far a computer algebra
verification of the generic Green Conjecture was feasible, the main problem
being the memory requirements in the syzygy computation.

The full Green conjecture is the following

Conjecture 5.1 (Green’s Conjecture [Gr84]). Let C be a smooth projective
curve of genus g ≥ 3 over a field of characteristic zero. The canonical ring
RC =

∑
nH

0(C, ω⊗n
C ) of C as an S = SymH0(C, ωC)-module has vanishing

Betti numbers βi,i+2 = 0 if and only if i < Cliff(C).

Here the Clifford index is defined as

Cliff(C) = min{degL− 2(h0(L)− 1) | h0(L), h1(L) ≥ 2}.

This conjecture generalizes the Noether-Petri-Babbage theorem which are
the cases i = 0 and i = 1 respectively. Its known for i = 2 in full generality,
see [Vo88, Sch91]. The direction Cliff(C) ≤ i ⇒ βi,i+2 6= 0 was established
by Green and Lazarsfeld in the appendix to [Gr84]

Green’s conjecture is known to be wrong over fields of finite characteristic.
The first cases are g = 7 in characteristic 2 and genus g = 9 in characteristic
3, see [Sch86, Sch03].

Frequently, Green’s Conjecture is formulated with the property Np of
Green-Lazarsfeld [GL85]: A subscheme X ⊂ Pr satisfies the property Np if
βij = 0 for all j ≥ i+2 and i ≤ p, in other words if the first p steps in the free
resolution are as simple as possible. An equivalent formulation of Green’s
Conjecture is that a canonical curve satisfies property Np iff p < Cliff(C).

If the coordinate ring SX for general X ∈ U ′ ⊂ H ⊂ Hilbp(t)(P
r) is pure,

then
Z = {X ∈ U ′ | SX does not satisfy MRC}

is a divisor, a so-called Koszul divisor, because in principle it can be computed
with the Koszul complex and Koszul cohomology groups.

For odd genus g = 2k − 1 Hirschowitz and Ramanan computed the class
of the corresponding divisor in Mg.

Theorem 5.2 (Hirschowitz-Ramanan [HR98]). If the generic Green conjec-
ture holds for odd genus g = 2k − 1, then the Koszul divisor

Z = {C ∈Mg | RC does not satisfy MRC } = (k − 1)M1
g,k

where M
1
g,k = {C ∈Mg | ∃ g

1
k on C} is the Brill-Noether divisor.

The proof is based on a divisor class computation of Z and M
1
g,k on a

partial compactification of Mg and the fact that Z − (k−1)M1
g,k is effective.

The coefficient k − 1 is explained by the fact that a curve with a g1k has
βk−1,k = βk−2,k ≥ k − 1 and equality holds for a general k-gonal curve of
genus g = 2k − 1 as a consequence of the Hirschowitz-Ramanan Theorem.

Based on the Hirschowitz-Ramanan and Voisin’s Theorem, Aprodu and
Farkas [AF09] established that Green’s Conjecture holds for smooth curves
on arbitrary K3 surfaces.

One can hope that there are various further interesting Koszul divisors.
To get divisors on Mg we consider the case that g = (r + 1)s, d = r(s + 1)
and ask whether a normal curve of degree d, genus g and speciality h1(L) = s
can have a pure resolution. This is a purely numerical condition on r and s.

Conjecture 5.2 (Farkas, [Fa06]). Let p ≥ 0 and s ≥ 2 be integers. Set
r = (s + 1)(p + 2) − 2, g = (r + 1)s and d = r(s + 1). A general smooth
normal curve C ⊂ Pr of genus g, degree d and speciality h1(OC(1)) = s has
a pure resolution, equivalently it satisfies Np.

If the conjecture is true, then the divisor

Zp,s = {C ∈Mg | ∃g
r
d which does not satisfy property Np}

gives counterexamples to the slope conjecture, see [AF08]. For small cases
Farkas verified the conjecture computationally in the spirit of the computa-
tions below.

Turning to non-special curves, we have

Conjecture 5.3 (Farkas, [FL10]). A general smooth normal curve C ⊂ Pr

of odd genus g = 2p+3 and degree d = 2g has a pure resolution, equivalently
satisfies Np. A general smooth normal curve of even genus g = 2p + 6 and
degree 2g − 2 has a pure resolution, equivalently satisfies Np.

For the rest of this section we return to the task to verify these conjectures
computationally in a few cases. The case of genus g = 8 and degree d = 14
has been established along the proof of Verra’s Theorem in Section 4. We
treat the case of genus g = 7 and degree d = 14. This case was settled in
[Fa09] using reducible curves. Here we use smooth curves.

Consider for g = 7 a random plane model of degree d = 7 with δ = 8
nodes.

i62: FF=ZZ/10007

i63: R=FF[x_0..x_2]

i64: g=7

i65: delta=binomial(6,2)-7

i66: J=randomDistinctPlanePoints(delta,R)

i67: betti res J

i68: betti (J2=saturate(J^2))

i69: C=ideal (gens J2*random(source gens J2,R^{-7}))

To find a general divisor of degree 2g = 14 on C we note that over a
large finite field geometrically irreducible varieties have always a lot of F-
rational points. We can frequently find points as one of the ideal theoretic
components of the intersection with a random complementary dimensional
linear subspace.

i70: decompose(C+ideal random(1,R))

i71: apply(decompose(C+ideal random(1,R)),c->degree c)

i72: time tally apply(1000,i->

apply(decompose(C+ideal random(1,R)),c->degree c))

-- used 9.62383 seconds

o72 = Tally{{1, 1, 1, 1, 1, 2} => 5}

{1, 1, 1, 1, 3} => 18

{1, 1, 1, 2, 2} => 21

{1, 1, 1, 4} => 47

{1, 1, 2, 3} => 84

{1, 1, 5} => 91

{1, 2, 2, 2} => 19

{1, 2, 4} => 120

{1, 3, 3} => 58

{1, 6} => 167

{2, 2, 3} => 40

{2, 5} => 94

{3, 4} => 77

{7} => 159

o72 : Tally

Thus the following function will provide points.

i73: randomFFRationalPoint=method()

i74: randomFFRationalPoint(Ideal):=I->(

--Input: I ideal of a projective variety X

--Output: ideal of a FF-rational point of X

R:=ring I;

if char R == 0 then error "expected a finite ground field";

if not class R === PolynomialRing then

error "expected an ideal in a polynomial ring";

n:=dim I-1;

if n==0 then error "expected a positive dimensional scheme";

trial:=1;

while (

H:=ideal random(R^1,R^{n:-1});

pts:=decompose (I+H);

pts1:=select(pts,pt-> degree pt==1 and dim pt ==1);

#pts1<1 ) do (trial=trial+1);

pts1_0)

i75: randomFFRationalPoint(C)

This allows to get a random effective divisor of degree 14 on C with all
points F-rational, for which we verify N2:

i76: points=apply(2*g,i->randomFFRationalPoint(C))

i77: D=intersect points -- effective divisor of degree 14

i78: degree D

o78 = 14

i79: DJ=intersect(D,J)

i80: degree DJ==degree D + degree J

o80 = true

i81: betti DJ

i82: H=ideal(gens DJ*random( source gens DJ, R^{-6}))+C

i83: E=(H:J2):D -- the residual divisor

i84: degree E +degree D + 2*degree J == 6*7

o84 = true

i85: L=mingens ideal ((gens truncate(6,intersect(E,J))) %C)

-- the complete linear series |D|

i86: RC=R/C

i87: S=FF[y_0..y_7]

i88: phi=map(RC,S,sub(L,RC))

i89: I=ideal mingens ker phi;-- C re-embedded

i90: (dim I,degree I, genus I) == (2,14,7)

i91: time betti res I

0 1 2 3 4 5 6

o91 = total: 1 14 28 56 70 36 7

0: 1 . . . . . .

1: . 14 28 . . . .

2: . . . 56 70 36 7

The computation above is the reduction mod p of the computation for a
curve together with a divisor over some open part of SpecOK of a number
field K. We can bound the degree [K : Q] ≤ 714 of the number field, a bound
on its discriminant seems out of reach.

Hence as before we can conclude that Farkas’ syzygy conjecture holds
for curves of genus 7 and degree 14 over fields of characteristic zero. Using
Mukai’s description of curves of genus 7 [Mu92], we could find an example
defined over Q. I chose to present the example above because I wanted to
illustrate the trick to get points in Mg,n from points in Mg over finite fields.

It is not difficult to use similar constructions to verify Farkas’ conjecture
for other small g.

It is known that the MRC is not satisfied in a number of important cases,
notably for non-special normal curves of genus g ≥ 4 and large degree [Gr84].

